The swirl effect
By RubyNL

The swirl effect is some really cool effect that I made from a small program from Jesse Regier aka solus. The original program can be found here: http://homepage.usask.ca/~jlr353/code/qbasic/swirly.bas. When I saw that program I found it a cool idea and I decided to work it out further. And, when I optimized it and made a moving swirl, I decided to make a tutorial for it because it was not hard to make at all.

The basic equations – Finding the angle of rotation
We need to rotate some points, but only the points that within the circle with radius R. 
Thus, we first need to know if the current point is in the circle, else you don’t have to rotate it at all. After that we can start to find the angle that the point needs to be rotated. So to check if the current point is in the circle: We first need to check in a loop for all points from –radius to radius if the current point is within the circle with radius R. We do that with a square root. (But that is so slow… Yes it is, but we’re going to optimize this later, so hold on!), like this:

FOR x = -r TO r

  FOR y = -r TO r

    distance = SQR(x ^ 2 + y ^ 2)

    IF distance =< r THEN

      ‘rotate the point

    END IF

  NEXT

NEXT

OK, that’s done. Now we need to have the equation to rotate the point. We have one clue: the further the point is from the centre the less it is rotated. There is an equation that is less when the distance from the centre is more(it is inversely proportional):

((r – distance) / r) 
Ok, the only thing we have to do is finding the angle of rotation, that’s really simple, just multiply the last equation by the amount of rotation. The amount of rotation is 
x * 2 * pi
Two times pi = 360 degrees in radians, so if x = 1, the swirl wraps around itself 1 time. And if x = 2, the swirl wraps itself around two times, and that’s also like that for each value of x. So to make a animating swirl, you just add some value to x, and if x reaches the max-wrapping-around (I use one for that).

So now we have(notice I replaced 2 * pi with 6.28, that’s two times pi):

angle! = x * 6.28 * ((r – distance) / r) 

So, remember that, we gonna need it later!
The basic equations – Rotating the point by angle!

Now we need to rotate the point, this is not so hard, you only need the inverse rotation formula:

rotatedx = x * COS(angle!) - y * SIN(angle!)

rotatedy = x * SIN(angle!) + y * COS(angle!)
The inverse rotation formula is needed when you want to show something rotated on the screen. The thought behind it is that you don’t rotate every point from the picture and show them all on the screen, but do the inverse: rotate every pixel on the screen and check on which pixel from the picture they ‘fall’. So, when you use all you’ve learned you should be able to make something like this(if you can’t, then look at it closely):

Notes: 

· I replaced the x from last equation by a!, to avoid confusion.

· I use a XOR background, by XORring rotatedx by rotatedy and use it as color, if you want a swirl over some picture you just have to replace the “rotatedx XOR rotatedy” by “texture(rotatedx, rotatedy)” or, depending on how you saved your original image, PEEK to it.
SCREEN 13  'Screen mode 13

DEFINT A-Z  'For speed

CONST r = 100

angleplus! = .05
DO

  a! = a! + angleplus!

  If a! => 1 OR a! =< -1 THEN angleplus! = -angleplus!

  FOR x = -r TO r

    FOR y = -r TO r

      distance = SQR(x ^ 2 + y ^ 2)

      IF distance =< r then

        angle! = a! * 6.28 * ((r - distance) / r)

        rotatedx = x * COS(angle!) - y * SIN(angle!)

        rotatedy = x * SIN(angle!) + y * COS(angle!)
        PSET (x,y), rotatedx XOR rotatedy

      END IF

    NEXT

  NEXT

LOOP

That brings us to another problem: the swirl is at the top left, you can see only ¼ of it! This is easy to fix, just add cx(center x) to x and cy(center y) to y, and cx to rotatedx and cy to rotatedy, all in the PSET.
So from:

PSET (x, y), rotatedx XOR rotatedy

To:

PSET (x + cx, y + cy), (rotatedx + cx) XOR (rotatedy + cy)

Now you can make your own moving swirl, but it’s still very slow because of the square root and the COS and SIN’s. So we have to make a few lookup tables.
The LUT’s
The first thing to optimize is the sines and cosines that are very slow. We can use LUT’s (Look Up Tables)! To make a COS/SIN LUT with fixed point math, we have to define how large it has to be. You can easily define the new range by doing a FOR…NEXT loop(in the example I used x as loop variable) from zero to range, and i.e. the SIN LUT be SIN(x / range * 2 * pi).  Like this, we first get our variable x in a 0-1 range, then by multiplying it by 2 * pi, we get it in the radians range(name of the angle that COS and SIN use). 

And, because we use fixed point math, we have to multiply it by a scale. So:
CONST scale = 128

CONST range = 255

CONST max = 255

DIM st(-range to range) as integer, ct(-range to range) as integer

pi! = ATN(1) * 4 or 22 / 7 or 3.14

FOR x = 0 TO range

  st(x) = SIN(x / max * 2 * pi!) * scale

  ct(x) = COS(x / max * 2 * pi!) * scale

NEXT

And, we need to divide the x and y coordinates by the scale again. Use \ (integer divide) for some real speed.

If range = max here, then the sin and cos tables loops around fully, if it’s less then not, and if range > max, then the table has some double parts in it(so it’s actually a waste of memory). I’ve used as much constants as I could, because they provide a little more speed, and, if somebody reads the code, he’ll notice that the values of constants can not change.

Here’s the program we have now, notice that a is an integer now:
SCREEN 13  'Screen mode 13

DEFINT A-Z  'For speed

CONST scale = 128

CONST range = 255

CONST max = 255

DIM st(-range to range) as integer, ct(-range to range) as integer

pi! = ATN(1) * 4

FOR x = 0 TO range

  st(x) = SIN(x / max * 2 * pi!) * scale

  ct(x) = COS(x / max * 2 * pi!) * scale

NEXT

CONST r = 100

CONST cx = 160

CONST cy = 100

angleplus = 1

DO

  a = a + angleplus

  If a => range OR a =< -range THEN angleplus = -angleplus

  FOR x = -r TO r

    FOR y = -r TO r

      distance = SQR(x ^ 2 + y ^ 2)

      IF distance =< r then

        angle = a * ((r - distance) / r)

        rotatedx = x * ct(angle) - y * st(angle)

        rotatedy = x * st(angle) + y * ct(angle)
        PSET (x + cx, y + cy), (rotatedx \ scale + cx) XOR (rotatedy \ scale + cy)

      END IF

    NEXT

  NEXT

LOOP

Distance LUT

Man, that’s nice! But there’s still a very slow square root left. We simply solve this with a distance lookup table. We don’t go from –r TO r in x and y, but from 0 TO r, because distance(x, y) = distance(-x,-y). So:

DIM distance(r, r)

FOR x = 0 TO r

  FOR y= 0 TO r

    Distance(x, y) = SQR(x ^ 2 + y ^ 2)
  NEXT

NEXT

And:

distance = distance(ABS(x),ABS(y))

The reason for ABS is that we need to have positive numbers, because we didn’t calculated for –x and –y, to save some memory. Use this instead of the square root in the main loop and it will execute MUCH faster!

A Moving Swirl

A moving swirl is no trick at all: don’t make cx and cy constants(!) and calculate a new value for them every time in the DO…LOOP part. For a swirl bouncing on the screen:
speedx = 5

speedy = 3

DO

  IF cx + speedx > 320 OR cx + speedx < 0 THEN speedx = -speedx

  IF cy + speedy > 200 OR cy + speedy < 0 THEN speedy = -speedy

  cx = cx + speedx

  cy = cy + speedy

  …

LOOP

Now you can see that the ‘previous’ swirl isn’t erased, so it looks shitty. What you need to do is check for every pixel if it’s in the swirl, and else erase it, this can be done by a simple loop like this:
FOR x = 0 TO 319

  FOR y = 0 TO 199

    xd = ABS(x – cx)

    yd = ABS(y – cy)

    IF xd < r AND yd < r THEN

      'Do the equation

    ELSE

      PSET (x, y), x XOR y

    END IF
  NEXT

NEXT

Of course, this needs to be in a loop again. And you need every variable like cx and cy again, and instead of what x and y were in the last equations, you use xd and yd. Further the ‘do the equation’ part is just the same.

Optimising it

You could optimise it even more and make tons of LUT’s for the integer divide part, but that’s not making it really faster. What should make it faster, is checking less pixels. We are now checking for each pixel on the screen, while it’s just used to look if any pixels are painted, but because the swirl moves too, not are painted over(so they still are rotated points). What should make it faster, is instead of painting each pixel over, just save the original image in a buffer and put it on the screen every frame. Then do the loop from –r to r again, but now check if the pixel you want to check if on the screen with a simple IF x => 0 AND y => 0 AND x < 320 AND y < 200 THEN … END IF.

Screen buffers for optimisation and reducing the flicker

When we really want to make this look neat, we need a screen buffer to reduce the flicker, and get just a little bit more speed. We’re going to need two screen buffers, but first, you need to know what’s a screen buffer.
Using great pictures instead of lame XOR, OR or AND generated textures
